

TSBDFT

柳瀬 章

平成 19 年 9 月 21 日

1 はじめに

バンド計算は通常逆格子空間（波数空間）の適当な代表点を選んで実行される。このとき選ばれる代表点の数は数十から数百が選ばれる。しかしこれでは金属的な物質でフェルミ面を作図したり、熱起電力などの伝導に関する量の計算に十分な密度ではない。たとえばこのホームページで紹介した TPERSP の説明に使われたフェルミ面の作図には 15 万点を超える波数でのエネルギー値が使用されている。ここで紹介するプログラム TSBDFT は、あたえられた対称性に基づいて、任意の波数ベクトルでのバンドエネルギーをあたえる解析的な関数を求め、それを用いて必要な密度でバンドエネルギーを計算する。TSBDFT では TSPACE の機能を使っている。ときどき説明に使われる専門用語に戸惑われる方は、たとえば、拙著「ブリルアンゾーンとは」(丸善)などを読んでいただきたい。また「空間群のプログラム TSPACE」の 3、4 章で説明している TSPACE が使用している実格子空間、逆格子空間での座標系、単位系を理解した上で、この説明を読んでいただきたい。TSBDFT では、TSPACE の上で使用するように設計されている。したがって「空間群のプログラム TSPACE」の記述と重複する部分が表れる。その場合には TSP:Pxxx のようにして引用する。

TSBDFT が使用する逆格子空間の代表点は、TSPACE が想定する逆格子空間の座標軸 a^*, b^*, c^* 軸に沿って、原点から、その方向の再近接の逆格子点までを等間隔に刻んだ 3 次元格子の上に採られていることを期待している。これらの点は、必ずしもブリルアンゾーンの中に含まれている必要はない。ただし、互いに同等な点を含んでいると誤動作の原因になる。当然なことであるが、3 次元の 3 方向で、間隔をできるだけ等しくとってある必要がある。例えば c 軸が長く c^* が短い結晶の場合は c^* 軸方向の刻みの数より a^* 軸方向の刻みを多くとって、互いの間隔が近くなるようにする必要がある。TSPACE にはサブルーチン TSKPGN が用意されているのでこの利用が便利である。TSKPGN の出力では、 $k = 0$ の Γ 点が入るようになってくるようになってくるが TSBDFT のためには、このようになってくるのが望ましい。

逆格子空間の代表点の選び方に、 a^*, b^*, c^* 軸に沿うのではなく、基本逆格子ベクトルに沿って刻む方法もあるが、TSBDFT のためには、この方法は適当ではない。最も分かりやすい例は底心直方格子の場合である。二つの a^*, b^* 面内の基本逆格子ベクトルは、長さが等しくどちらも a^*, b^* 軸の間になっている。したがって基本逆格子ベクトルに沿って刻むと、 a^*, b^* の長さに関係なく a^*, b^* 方向の刻みが同じになってしまう。この二つの軸の長さの違いが 2 倍以上の物質が実際に数多く存在している。このような代表点の取り方は TSBDFT のためだけでなく、密度汎関数理論に則った、電荷密度の計算のためにも望ましくはない。

基本格子ベクトルや基本逆格子ベクトルを使う議論は、式の上では簡潔になり便利であるが、我々のあまり得意でない 3 次元感覚をさらに狂わせてしまう欠点を持っている。基本格子ベクトルで表した係数には、そのベクトルが持っている対称性が直接には反映されていないからである。これが、実用的なプログラムを目指す TSPACE が、 a, b, c, a^*, b^*, c^* を基準にした、座標系に固執している理由の一つである。

以下のプログラムの説明では、ソースプログラムをアルゴリズムの鍵になる部分以外は、一々引用しない

ことが多いので各自でソースを手元に持ちながら読んでいただきたい。

2 内挿関数

バンドエネルギー $E(k)$ は波数空間での周期性がある。

$$E(k + G) = E(k) \quad (1)$$

ここで G は任意の逆格子ベクトルである。任意の実格子ベクトルを T_n として $\exp(iT_n \cdot k)$ はこの周期性を持っている。

さらに、結晶が回転の対称性を持つ場合には $E(k)$ もやはり同じ回転の対称性を持つ。

$$E(k) = E(\alpha k) \quad (2)$$

らせん、映進のように、結晶の周期でない並進を伴っている場合でも同じように成り立つ。

空間群 G を、その並進群 T を法として剰余類に

$$G = (\epsilon|0)T + (\alpha|u_\alpha)T + (\beta|u_\beta)T + \dots \quad (3)$$

と分類すると $\epsilon, \alpha, \beta, \dots$ は一つの結晶点群 G_P になる。そこで

$$\sum_{\alpha \in G_P} \exp(i\alpha T_n \cdot k) \quad (4)$$

とすれば、

$$\sum_{\alpha \in G_P} \exp(i\alpha T_n \cdot \beta k) = \sum_{\alpha \in G_P} \exp(i\beta^{-1}\alpha T_n \cdot k) \quad (5)$$

だから、回転の対称性を持った関数になる（組み換え定理）TSP:p9。

バンドエネルギー $E(k)$ にはもう一つの重要な対称性がある。それは時間反転の対称性に起因している。結晶が空間反転の対称性を持っていれば、当然

$$E(k) = E(-k) \quad (6)$$

となるが、たとえ空間反転の対称性がない場合でも、このことを時間反転の対称性が導く。スピン軌道相互作用を含めない議論では、これで議論を終わりにしてよい。強磁性状態などで、スピンの縮退が解けていて時間反転の対称性が壊れている場合でも、空間座標とスピン座標が互いに関係していないので、空間座標の部分では時間反転の対称性は保たれているからである。もちろん時間反転で生じる、ペアリング、ダブリングの縮退度の変化は当然考慮にいれてのことである。

スピン軌道相互作用を含めると議論が次のように変わる。磁性状態では時間反転の対称性は壊れているので、ここでは扱わない。空間反転がある結晶では、時間反転と空間反転の両方が $E(k) = E(-k)$ となるように働く。このとき時間反転はスピンの向きを反転させるのに対し、空間反転はスピンの向きを変えないので、結果的にスピンの自由度に起因する二重縮退がすべての k で残ることになる。空間反転を持たない結晶では、時間反転だけが $E(k) = E(-k)$ となるように働くため、一般の k ではこの二重縮退が残らない。

とにかく $E(k) = E(-k)$ は保証されているので、

$$\sum_{\alpha \in G_P} \cos(\alpha T_n \cdot k) \quad (7)$$

は $E(k)$ を表す基底として使用できる。

しかし、実際の $E(k)$ は k の関数として連続ではあるが、その微分は連続ではない。(7) は微分が連続であるから、このことを表現できない欠点を持っている。微分が不連続になるのはつぎの2つの場合が挙げられる

1. 対称軸、対称面での偶然縮退
2. 一般の点から対称面、対称面から対称軸、対称軸から対称点などの対称が変わる点での接続

これに対処する方法はいろいろ考えられるが、ここで紹介する TSBDFE では、このことの手当はしていない。汎用的に使用できるようにすることを優先したからである。この内挿法を使えば、TPERSP の応用でフェルミ面を描くことが、ほとんどすべての結晶で可能になる。描かれたフェルミ面に不自然な点があれば、代表点の数を増やして変化がないかどうか確かめる必要がある。特に偶然縮退がフェルミ面の近くで起きていれば、(7) が、偶然縮退の不自然な変化をカバーしていること確かめるなどの注意を払うべきである。AYBAND を使って該当する箇所を JOPT=1 にしてプロットしてみるのもおすすめである。AYBAND では上に挙げた二つの問題点はカバーされている。AYBAND の JOPT=1 の図は、使われている代表点の位置が図面に現れるので、特異な形のフェルミ面が本物かどうかの判定に役に立つ。とにかく、もっともらしく見える結果の真偽のほどを保証するのは使用者個人の責任である。

3 プログラムの流れ

プログラム 3.1 は TSBDFE のメインプログラムの全体である。プログラム 3.2 は最初の SUBROUTINE TSPPRP で、プログラム 3.3 は二番目の SUBROUTINE FITPRM である。ENRRED、SSSLAT、FNNMAT、FUEINT については、節を変えて説明する。

プログラム 3.1

```
C-----1-----2-----3-----4-----5-----6-----7-----8
C   Main for fitting of Bands energies and components
      CALL TSPPRP
      WRITE(6,*) ' End of TSPPRP'
      CALL FITPRM
      WRITE(6,*) ' End of FITPRM'
      CALL ENRRED
      WRITE(6,*) ' End of ENRRED'
      CALL SSSLAT
      WRITE(6,*) ' End of SSSLAT'
      CALL FNNMAT
      WRITE(6,*) ' End of FNNMAT'
      CALL FUEINT
      WRITE(6,*) ' End of FUEINT'
      STOP
      END
```

プログラム 3.2

```
C-----1-----2-----3-----4-----5-----6-----7-----8
      SUBROUTINE TSPPRP
C*****
C PREPERATION TO CALL TSPACE
C*****
      IMPLICIT REAL*8 (A-H,O-Z)
      COMMON/SCL/NATOM,NKAT,IPR,ISO,E0,DE,NNEE,NELEC,ENEND
      DIMENSION JB(2,3)
      READ(1,*)
      READ(1,*) IL,NGEN,INV
      CALL TSPACE(IL)
      DO 1 I=1,NGEN
```

```

READ(1,*) JA,((JB(J,K),J=1,2),K=1,3)
CALL TSGENR(JA,JB)
1 CONTINUE
CALL TSPGRP(INV)
IF(IPR.GE.3) CALL TSPGDS
READ(1,*) A,B,C
READ(1,*) CA,CB,CC
CALL TSLATC(A,B,C,CA,CB,CC)
READ(1,*) NATOM,NKAT
RETURN
END

```

fort.1 からデータを読み込んで格子の型 IL と生成元をあたえて、空間群を決定したあと、格子定数をあたえている。最後の READ(1,*) NATOM,NKAT は、当面必要がないが、4 節で説明する s, p, d, f 成分の読み込み用である。このプログラムは TSP:p71 p73 の例 5.2,5.3 とほとんど同じである。TSPACE が使用している空間群、格子定数のあたえ方の詳細はこの部分を参照されたい。

プログラム 3.3

```

C-----1-----2-----3-----4-----5-----6-----7-----8
SUBROUTINE FITPRM
IMPLICIT REAL*8 (A-H,O-Z)
INCLUDE 'prmfrit.f'
COMMON/SPG2/IL,NG,IG(48),JV(2,3,48)
COMMON/NXYZ/NX,NY,NZ,N64,JS
CHARACTER*4 CHARA4
CHARACTER*10 MAGNET
C
READ(5,150) MAGNET
150 format(A10)
CHARA4=MAGNET
IF(CHARA4.EQ.'NONM') ISO=1
IF(CHARA4.EQ.'MAGN') ISO=2
IF(CHARA4.EQ.'SPIN') ISO=3
NDFAC=1
IF(ISO.EQ.3) THEN
DO 10 I=1,NG
IF(IL.GE.1.AND.IG(I).EQ.25
& .OR.IL.LE.0.AND.IG(I).EQ.13) NDFAC=2
10 CONTINUE
END IF
IF(ISO.EQ.3.AND.NDFAC.EQ.1) ISO=4
write(6,*) 'NDFAC=',NDFAC,'ISO CODE=',ISO
READ(5,*) NELEC,NB1,NB2
READ(5,*) AM
READ(5,*) IPR
READ(5,*) NX,NY,NZ,N64,JS
NBLOCK=1
RETURN
END

```

このサブルーチンは標準入力 (fort.5) から読み込んで、内挿プログラムの仕様を決める。最初に CHARACTER*10 変数 MAGNET に非磁性 (NONMAGNETIC)、強磁性 (MAGNETIC)、SPIN 軌道相互作用を含めた計算 (SPIN-ORBIT) を読み込んで、入力されるバンド計算がこのいずれかであるかを指定する。2 節で述べたようにスピン軌道相互作用を含めた場合には、空間反転のあるなしが $E(k)$ の縮退に影響するので、ここで、それがあつた場合を ISO=3、ない場合を ISO=4 として区別している。

つぎの READ 文の NELEC は計算されたバンドに含まれるはずの電子数をあたえるが、TSBDFIT では結果として使用しないので、なにがあたえられても結果に影響しない。バンド計算のプログラムが原子番号等から自動的にあたえてくれているのを、使用者が再認識するようにこの入力を消さずに置いてある。内挿は NB1 番目のバンドから NB2 番目のバンドまで実行される。常磁性の物質ではフェルミ面があるバンドをカバーするように選ばばよい。半導体や絶縁体ではバンドギャップを挟む適当な数のバンドを選ぶ。

AM は SSSLAT で用意する格子ベクトルの大きさの範囲をあたえる。IPR はプログラムの出力制御用である。この TSBDFE はあたえられた k -点の数が増加すると、必要なベースの数が増え、計算時間も長くなる。IPR の値としては 0,3,4,5 が現在のバージョンには使われている。適当に選んで計算の進行をトレースすることができる。NX,NY,NZ,N64,JS については、それらが使用されるサブルーチン FUEINT の説明の中でふれる。

4 バンド計算の結果の読みとり:ENRRED

プログラム 4.1 はサブルーチン ENERED の中で、バンド計算の結果が出力されているファイルから読み込んでいる部分の抜粋である。われわれが使っている FLAPW の出力に合わせてある。もちろん他の方法で計算されたものでも、この形式に合わせてあれば TSBDFE の入力として使用できる。40 READ(2,150,END=41) の FORTAN の手法でファイルの終わりを検出しているので、データが何組あるかは、指定しなくてもよい。各 \vec{k} 毎に、またその \vec{k} での既約表現ごとにエネルギーの低いものから順に並んでいることを期待している。

プログラム 4.1

```

40 READ(2,150,END=41) KK,(KX(J),J=1,3),IC,IUD,MRR,MRN,NDEG,NST,NEIG
  IF(NEIG.GT.MAXEIG) THEN
    WRITE(6,*) ' MAXEIG=',MAXEIG,' NEIG=',NEIG
    STOP
  END IF
  IF(NEIG.GT.0) READ(2,*) (EIG(J),J=1,NEIG)
150 FORMAT(I3,2X,3I3,1X,2I3,2X,A2,I2,3I3)

```

READ 文の並びに現れた変数の意味は

- KK — バンド計算のときに \vec{k} につけられた通し番号、TSBDFE では読み捨てる。
- KX(3) — \vec{k} の座標の分子。
- IC — \vec{k} の座標の分母。
- IUD — スピンの UP(1) or DOWN(2)、パラのときには 1。
- MRR — \vec{k} の名前、TSBDFE では用いない。
- MRN — 既約表現の番号、TSBDFE では使用しない。
- NDEG — 縮退度、TSBDFE ではこれを元に既約表現ごとにあたえられたエネルギーを、大きさの順番に並べ直している。なお、プログラム 3.3 で ISO=3 とした場合（スピン軌道相互作用を含めていて、空間反転を持つ場合）は NDEG は偶数になっているので、確かめた上で NDEG を半分にしていく。
- NST — \vec{k} の星の数、TSBDFE では用いない。
- NEIG — その既約表現に属する、状態の数。

である。変数の型は FORMAT から分かるように MRR 以外はすべて整数型である。MRR は 2 文字の文字型変数であるが、上記のように用いないのでなが入っていてもよい。もちろんスペースでもよい。

入力データ 4.1

```
45 4 4 4 8 1 R 1 4 1 5
-1.90115 -.46922 .19579 .53289 .90794
.3088 .1071 .0047 .0005
.0261 .2785 .0037 .0005
.0034 .2197 .0018 .0002
.0343 .0682 .0015 .0001
.0061 .0432 .0038 .0002
45 4 4 4 8 1 R 2 4 1 4
-.73549 -.57574 .45021 .88001
.2398 .0529 .0002 .0000
.0059 .2323 .0024 .0004
.0283 .0528 .0014 .0001
.0050 .0064 .0014 .0000
45 4 4 4 8 1 R 5 4 1 2
.22001 .93867
.0000 .2484 .0020 .0003
.0000 .0528 .0028 .0001
45 4 4 4 8 1 R 6 4 1 2
-.56316 1.06384
.0000 .2631 .0028 .0004
.0000 .0427 .0015 .0001
```

入力データ 4.1 は窒素の分子結晶に対する、バンド計算結果の R 点の部分である。ここでの k 点群は位数が 24 の T_h で 2 次元ばかり 6 個の既約表現を持っている。 R_1, R_3 と R_2, R_4 のペアがそれぞれ時間反転でペアリングをして 4 重縮退になる。また R_5 と R_6 は時間反転でダブリングをしている。TSBDFT は時間反転でペアリング、ダブリングが起きた場合にはこの例のように扱われていることを期待している。

1. ペアリングでは番号の若い方の既約表現の縮退度を 2 倍して、そのペアの相手の既約表現は出力に入れない。
2. ダブリングでは縮退度を 2 倍して、同じエネルギーを二つ並べて出力することはない。

この入力データにはエネルギーに続いて、そのエネルギーを持つ状態の性格を示す数値がそれぞれ 4 個づつついているが、TSBDFT はこの部分も読みこんで格納している。これは、各状態の、それぞれの原子球内での s, p, d, f 成分である。将来この性格を示す数値も合わせて内挿できるようにする予定である。

サブルーチン ENRRED は fort.2 からの入力データを整理して

プログラム 4.2

```
COMMON/ENR/KXM(3,MAXKPT),ICM(MAXKPT),IUDM(MAXKPT)
& ,NEIGM(MAXKPT),EIGM(MAXEIG,MAXKPT),NKPINT,NBAND,NB1,NB2
```

に格納した後、バンド数 NBAND を各 k でのエネルギー値の数の最小値にして、もし NB2 がこれを超えていれば、NB2=NBAND としている。このサブルーチンは最後に NBAND,NB1,NB2 を標準出力に出している。NKPINT は、代表点の数であるが、強磁性状態の場合には UP と DOWN を別のデータとして扱っているので、実際の代表点の数の 2 倍になっている。

5 内挿基底関数の準備:SSSLAT

SSSLAT では、TSPACE が内部で使用している隠しサブルーチンや内部変数を直接使用している。プログラムの流れの説明のなかで触れていくことにする。

サブルーチン SSSLAT の目的は式 (7) の基底関数を組み上げることである。結果が次の COMMON 領域に格納される。

プログラム 5.1

```
COMMON/LATVC/LATOM(3,MAXLP),ALM(MAXLP),MP(MAXLP),NLAT,NSHEL,ICD
```

最後の ICD は格子ベクトルの共通分母で、三方格子なら 3、底心、面心、体心格子なら 2 で、他の単純格子なら 1 になっている。

NSHEL は用意された基底関数の数、NLAT はそれらに含まれる格子ベクトルの総数である。

LATOM(3,MAXLP) には、NLAT 個の格子ベクトルが並べてあり、その大きさが ALM(MAXLP) である。この中には $T_n = -T_m$ の関係になるような格子ベクトルは含まれないようにしてある。MP(MAXLP) は、基底関数を構成している最後の格子ベクトルの LATOM 上の番号が入っている。

このサブルーチンの流れは次のようである。

1. まず格子の型によって、TSP:p62~63 の (5.2)~(5.6) にしたがって基本格子ベクトルをあたえたのち、その整数倍の線形結合で順次格子ベクトルを作って、LATOM に格納していく。このとき大きさが FITPRM であたえた AM を超えるものを取り除きながら、大きさの順番にソートしておく。格子ベクトルの大きさの計算には、TSPACE の隠しサブルーチン ZZZY43(V,DC,S) が使われている。TSP:(5.2)~(5.6) は基本格子ベクトルを a, b, c を基準にした座標系であたえているので、使用している座標系はやはり、 a, b, c を基準にしたものである。

基本格子ベクトルの整数倍で格子ベクトルを生成するアルゴリズムはごく自然に見えるが、実際的には問題を含んでいる。たとえば、三方格子で、 c 軸が長い場合は、基本格子ベクトルは互いに接近している。したがって c 軸に垂直な方向の格子ベクトルを表すには、絶対値が大きな整数が必要になる。 c 軸が短いときには逆に基本格子ベクトルは互いに開いており、 c 軸の方向の格子ベクトルを表すのに大きな整数を必要とする。同様なことは格子定数が互いに大きく異なる体心直方格子のような場合にもおきる。サブルーチン SSSLAT では探索する整数の範囲を AM を基本格子ベクトルの大きさと割った値の 5 倍と一応大きめに設定している。しかしこれでも不十分なこともありうるので、その場合は 'STOP AT 26' とコメントを出してプログラムを止めるようにしてある。これは、ある格子ベクトルを種にして、それと回転の対称性で結びつく格子ベクトルを探している箇所が該当する格子ベクトルが見つからない場合に相当している。 a, b, c を基底にすると、 $1/2$ とか $1/3$ が現れて泥臭く見えるかも知れないが、このほうが明らかにわれわれの直感と結びついている。

2. 次のステップの最初に大きさの等しいものごとにグループに分ける。

```
DO 52 II=2,NLAT ~ 52 CONTINUE
```

このグループの数が MSHEL で、境目が ML(MAXLP) である。

3. 次にこの大きさの等しい格子ベクトルを、(7) 式を構成するように、さらにグループ分けする。

```
DO 81 IE=2,MSHEL ~ 81 CONTINUE
```

の長い手続きになる。(7) 式を構成する格子ベクトルは、基底の間で重なることはない。配列 IFLA(MAXLP) は、各格子ベクトルにつけた目印であり、同じ基底 (7) 式に含まれる格子ベクトルのそれは、互いに等しくなるようにしてある。

まだ使われていない格子ベクトルの IFLA は 0 にしてあるので、それを見つけてその番号を ISAR とする。DO 72 JFLA=1,50 に現れた、50 は同じ大きさになる格子ベクトルを (7) 式を構成するようにグループ分けしたときの最大の組数を大きめに見積もった値である。FCC で O_h の点群の場合で 3281 個の代表点に対応して 10302 組のベースを作った場合に 27 組が表れたのが最大である。次節で述べるように用意すべきベースの数は代表点の数の 2 倍から 3 倍が必要である。全部使われておれば、73 CONTINUE の外に出る。この場合は DO 72 のループの外に出てもよいが、GO TO 72 としているのは、単に文番号の数を節約しているだけである。

プログラム 5.2

```
DO 72 JFLA=1,12
```

```

      IF(JFLA.EQ.1) THEN
      ISAR=ISTA
      ELSE
      DO 73 I1=ISTA,IEND
      IF(IFLA(I1).EQ.0) GO TO 74
73  CONTINUE
      GO TO 72
74  ISAR=I1
      END IF

```

4. 上で決めた、たねの格子ベクトルを L ATF(3) として、これを空間群の回転要素で回転して、L ATT(3) とする。回転要素のコード番号は TSPACE の COMMON/SPG2/ の IG(48) に格納されていて、その数は NG 個である。また、このコード番号の回転操作は COMMON/SPG1/ の IT(3,48) に入っている。TSP:P7~11 のコード表で、X,Y,Z,W としてあるところが IT では、それぞれ 1,2,3,4 になっている。またたとえば、-X は -1 に -W は -4 のようにしてある。

L ATT(3) か -L ATT(3) は当然どこかにあるはずだから、見つけて JFLA を目印につけておく。もしみつからなければ、異常だから STOP が働く。

プログラム 5.3

```

      L ATF(1)=L ATOM(1, ISAR)
      L ATF(2)=L ATOM(2, ISAR)
      L ATF(3)=L ATOM(3, ISAR)
      IFLA(ISAR)=JFLA
      DO 22 I1=2,NG
      DO 23 K=1,3
      IF(IABS(IT(K,IG(I1))).EQ.4) THEN
      L ATT(K)=L ATF(1)-L ATF(2)
      IF(IT(K,IG(I1)).LT.0) L ATT(K)=-L ATT(K)
      ELSE
      M=IABS(IT(K,IG(I1)))
      L ATT(K)=L ATF(M)
      IF(IT(K,IG(I1)).LT.0) L ATT(K)=-L ATT(K)
      ENDIF
23  CONTINUE
      DO 26 ID=1,2
      DO 24 I2=ISAR,IEND
      IF(IFLA(I2).NE.0.AND.IFLA(I2).LT.JFLA) GO TO 24
      DO 25 K=1,3
      IF(L ATT(K).NE.L ATOM(K,I2)) GO TO 24
25  CONTINUE
      IF(IFLA(I2).EQ.0) IFLA(I2)=JFLA
      GO TO 22
24  CONTINUE
      L ATT(1)=-L ATT(1)
      L ATT(2)=-L ATT(2)
      L ATT(3)=-L ATT(3)
26  CONTINUE
      WRITE(6,*) IE,I1,ISTA,IEND,ISAR,L ATF,L ATT,' STOP AT 26'
      STOP
22  CONTINUE
72  CONTINUE

```

5. 最後の手続きは同じ目印のものを集めることで、また DO 78 JFLA=1,50 のループで行っている。

新しい結晶で使用するときに、IPR を 4 にして一度出力を見ておくのも無駄でないかもしれない。

6 内挿関数の係数を決定:FNNMAT

サブルーチンの最初に、 k 空間での真の代表点の数を NFSHEL としている。NKPINT は、代表点の数であるが、強磁性状態の場合には UP と DOWN を別のデータとして扱っているので、実際の代表点の数の 2 倍になっている。4 節の ENRRED であたえられている、逆格子空間の代表点 k に対して、5 節で求めた基底関数

$$\sum_{\alpha \in G_P} \cos(\alpha T_n \cdot k) \quad (8)$$

の値を計算していく。基底関数を構成する格子ベクトルの組は、最後の番号だけが MP(J) に登録されていて、最初の番号が登録されていないので、少し工夫が必要であるが、ソースを見れば分かると考える。

プログラム 6.1

```
DO 15 JJ=1,JP
  DO 16 K=1,NFSHEL
    IF(DABS(FT(K)-F(K,JJ)).GT.1.0D-5) GO TO 15
  16 CONTINUE
C      WRITE(6,*) ' COIN ',JJ,J,JP,NFSHEL
      GO TO 19
  15 CONTINUE
```

プログラム 6.1 で示した部分が、TSBDFT が使っている一つの重要なアルゴリズムである。

DO 12 J=1,NSHEL (NSHEL は基底の総数) で基底関数の値を一つずつ計算していくのであるが、計算された結果が、すべての代表点の k に対して等しくなっているものがあるので、それを取り除いている。(8) であたえられる基底関数は、互いに同じ格子ベクトルを全く含んでいないのだから、当然互いに独立である。しかし離散的な k 空間の代表点での値どうしはかならずしも独立ではない。大きさの小さい格子ベクトルから順に行列の各列を計算していく時に新しく計算されたものが、既存の列の線形結合であるかどうかの確かめが必要になる。ところが実際に計算してみると、この簡単な手続き、つまりすべての k に対して同じになるものを除く手続きだけで独立な NFSHEL 個の基底関数を選び出せることが判明した。しかもちょうど NFSHEL 個である。選び出された基底の番号が JPMAT(MAXKPT) に登録される。IPR を 4 から 5 にすると、小さな格子ベクトルのベースはすべて有効であるが、格子ベクトルが大きくなるにつれて無効のもの数が増えて、さんざん探した末に最後のベースが見つかるようすが再現される。

以上で NFSHEL 次元の正方行列ができるので、次は一次方程式を解いて係数を求めることになる。F と FF と同じ行列を用意しているのは、汎用の一次方程式のサブルーチンが元の行列を変更してしまうからである。あとで求められた係数を代入してあたえたエネルギー値に等しくなっているかを確認している。

ここで使用している MLUDIS は NUMERICAL RECIPES の LUDCMP を中で CALL している。これで LU 分解しておいて、各 k 点のエネルギー値をあたえて、それぞれのバンドの係数を決定していく。MLULEQ は、おなじパッケージの LUBKSB を CALL している。このところは、使い慣れたパッケージを使えばよいが、数百次元以上の行列を扱う必要があるので、実績と信用のあるパッケージを使う必要がある。

決定された係数 BB(J) を代入してあたえたエネルギー値が 10^{-13} 以内で再現されていることを確かめて、B(MAXKPT,MAXEIG,2) に格納して次のサブルーチン FUEINT に引き渡す。

ここではエネルギー値を方程式の左辺に入れて、エネルギー値を求める係数を決めているが、ここで、ENRRED で用意されている、s,p,d,f 成分を左辺にすれば、これらの成分値を求める係数が決定できる。

7 詳細な刻みでのバンドエネルギーの計算:FUEINT

6節の FNNMAT までの手続きは、すべての空間群で格子定数に無関係に使用でできるようになっている。しかしこの節以降で説明するサブルーチン、プログラムは空間群ごとに使い分ける必要がある。

この節では、立方格子、正方格子、直方格子で共通に使用できるものを紹介する。単純、面心、体心、底心格子を共通に扱う関係で、結果を格納する領域と、計算手続きの両方に数倍の無駄をするルーチンになっているが、最近の計算機事情を考慮して、プログラムの構造を単純にするほうに重点をおいたためである。このパッケージに入れてある SUBROUTINE FUEINT は六方格子にも対応するようにしてある。この部分は次の8節で説明する。

このサブルーチンに3節の FITPRM から引き渡される変数は

プログラム 7.1

```
COMMON/NXYZ/NX,NY,NZ,N64,NBLOCK,JS
```

の中に格納されている。NX,NY,NZ はそれぞれ a^*, b^*, c^* 軸の刻み数で、N64 はそれらの最小公倍数である。NBLOCK は今回のバージョンでは使わない。JS については表1に示す。

表 1: 結晶点群と JS の値

No	Point group	JS
1	D_{2h}, D_2, C_{2v}	1
2	C_{4h}, C_4, S_4	1
3	D_{4h}, D_{2d}, C_{4v}	2
4	T_h, T	1
5	O_h, O, T_d	3
6	$D_{6h}, C_{6v}, D_6, D_{3h}$	1
7	D_{3d}, D_3, C_{3v}	1

7.1 領域の設定

空間群に、空間反転の対称性がなくても $E(k) = E(-k)$ は時間反転が保証するので、 $E(k)$ の値を用意する k の範囲を決めるためには、空間群に空間反転を補って考えてもよい。表1に各行に並べられた点群の対称性を持つ結晶でははその行の最初にある点群として扱ってよいことになる。はじめの5個の点群は2番の C_{4h} を除き

$$k_x = 0, \quad k_y = 0, \quad k_z = 0$$

の対称面を持っているので、負の値の k -座標を持つ $E(k)$ はその座標の絶対値の座標での E と等しくなっている。したがって第一象限に限ってエネルギー値を用意しておけばよいことになる。 C_{4h} は $k_z = 0$ の対称面は持っているが $k_x = 0, k_y = 0$ の対称面を持っていないので、絶対値をとって第一象限に移すことはできないが、

$$k_x < 0, k_y < 0 \quad \longrightarrow \quad k_x \leftarrow -k_x, k_y \leftarrow -k_y \quad (9)$$

$$k_x < 0, k_y > 0 \quad \longrightarrow \quad k_x \leftarrow k_y, k_y \leftarrow -k_x \quad (10)$$

$$k_x > 0, k_y < 0 \quad \longrightarrow \quad k_x \leftarrow -k_y, k_y \leftarrow -k_z \quad (11)$$

の手続きで第一象限にもってこることができる。10 節で説明するフェルミ面作図用の関数サブルーチン FUNC にはこのための手当はしていない。動作テストができる実例があれば、簡単に手当ができると考えている。表 1 に含まれている D_{6h}, D_{3d} については次節で説明する。

D_{4h} では $k_x = k_y$ の対称面があるので、 $k_x \geq k_y$ の領域に制限できるので $JS = 2$ とする。 O_h では、さらに $k_x \geq k_y \geq k_z$ と領域をせまくできるので、 $JS = 3$ とする。 T_h では $JS = 1$ としている。これでは必要な領域の 3 倍になっているが、簡単のために無駄を承知でこうしている。

TSBDFT では、この JS の値は利用者が自分であたえるようになっている。表 1 にしたがって決めればよいのだが、上の説明を理解した上で利用されることを希望する。JS=1 が直方、JS=2 が正方、JS=3 が立方対称と単純になっていないことを確かめ、その理由をなっとくした上で利用されたい。

単純格子ならば、 k - 空間の周期は a^*, b^*, c^* である。したがって

$$0 \leq k_x \leq 0.5a^*, \quad 0 \leq k_y \leq 0.5b^*, \quad 0 \leq k_z \leq 0.5c^* \quad (12)$$

の範囲のエネルギー値があれば、この範囲を出る k に対しては、 a^*, b^*, c^* を引く操作と、絶対値をとる操作の組み合わせでこの範囲にいれることができる。(12) が D_{2h} の場合の既約なブリルアンゾーンになっている。

底心格子では c^* は k - 空間の周期であるが、 a^*, b^* は周期ではなく $2a^*, 2b^*$ がこの方向の周期になる。そこで

$$0 \leq k_x \leq a^*, \quad 0 \leq k_y \leq b^*, \quad 0 \leq k_z \leq 0.5c^* \quad (13)$$

の範囲をとることにする。この範囲は既約なブリルアンゾーンの 2 倍の体積になっている。しかし、例えば

$$0 \leq k_x \leq 0.5a^*, \quad 0 \leq k_y \leq b^*, \quad 0 \leq k_z \leq 0.5c^* \quad (14)$$

の範囲にして、既約なブリルアンゾーンと同体積にすることは可能である。しかし、10 節で説明するフェルミ面を描くための関数サブルーチン FUNC に

$$0.5a \leq k_x \leq a^*, \quad 0 \leq k_y \leq b^* \quad (15)$$

の範囲にある k -点を

$$0 \leq k_x \leq 0.5a^*, \quad 0 \leq k_y \leq b^* \quad (16)$$

の範囲に移し替える手続き

$$k_x \rightarrow a^* - k_x \quad (17)$$

$$k_y \rightarrow b^* - k_y \quad (18)$$

を入れておく必要が生じる。

面心格子や体心格子では $2a^*, 2b^*, 2c^*$ がそれぞれの方向での周期になる。そこで

$$0 \leq k_x \leq a^*, \quad 0 \leq k_y \leq b^*, \quad 0 \leq k_z \leq c^* \quad (19)$$

の範囲をとることにする。この範囲は既約なブリルアンゾーンの面心で 2 倍、体心で 4 倍である。面心直方、体心直方、体心正方（面心正方）格子では上の底心直方格子の場合と似た手続きで既約なブリルアンゾーンと同体積の領域に還元することが可能である。また立方格子の場合ではいくつかのブロックに分割することで、既約なブリルアンゾーンと同体積にすることは可能である。さきに説明を飛ばした NBLOCK はこのことに対応するように残してある。

今回のバージョンで使われる領域は T_h の体心格子で、既約なブリルアンゾーンの最高 1.2 倍になっている。もちろんこれらに対応することは可能であるが、個別のケースでそれぞれに対応するほうが実際的である。このバージョンではいろんな場合を一つのプログラムでカバーすることに重点を置いて、無駄の方には目をつぶっている。

FUEINT の中では、ここで述べた範囲は P(3) の中に記録されて、10 で述べる FUNC 用に fort.21 に出力される。またこの範囲を刻んでいる刻みの数は N(3) で、やはり fort.21 に出力される。

7.2 エネルギー値の計算

計算の中心は (7) の

$$\sum_{\alpha \in G_P} \cos(\alpha T_n \cdot k) = \sum_m \cos(T_m \cdot k) \quad (20)$$

の基底関数の計算である。ここでの k は、入力であたえて最小公倍数 N64 を N_L として

$$k = 2\pi(I_x/N_L)a^* + (I_y/N_L)b^* + (I_z/N_L)c^* \quad (21)$$

と整数 I_x, I_y, I_z で一般に表すことができる。

一方 $\alpha T_n = T_m$ は格子ベクトルであるから、格子の型で 1,2,3 と値を変える共通分母 C_d を用いて

$$T_m = (L_x/C_d)a + (L_y/C_d)b + (L_z/C_d)c \quad (22)$$

とやはり整数 L_x, L_y, L_z で表すことができる。したがって (20) の各項は

$$\cos(T_m \cdot k) = \cos(2\pi(I_x L_x + I_y L_y + I_z L_z)/(N_L C_d)) = \cos(2\pi(I_{kT}/(N_L C_d))) \quad (23)$$

と整数 I_{kT} を使って表される。

関数 \cos は偶関数で、周期が 2π 、さらに

$$\cos(x) = -\cos(\pi - x) = -\cos(\pi + x) = \cos(2\pi - x) \quad (24)$$

だから、0 から $\pi/2$ の区間を $N_L C_d/4 = N_q$ 分割した点で、 \cos の計算をしておけば必要なものは尽きることになる。FUEINT では、(24) を使って 0 から 2π の間の \cos の値を CRK(0:600) に用意している。プログラムの中では $N_L C_d = KRMESH$ 、 $N_q = MESH$ となっている。TSBDFT で、刻みの数 NX,NY,NZ に加えてそれらの最小公倍数 N64 を入力として要求しているのは、刻み数を選ぶときになるべく最小公倍数が小さくなることを使用者に意識してもらうためである。

これで準備が完了したので DO 21,DO 22,DO 23 でそれぞれ a^*, b^*, c^* 方向を刻んで k を定める。このとき JS の指定にしたがって、IY の上限 IYMM を IX にしたり、IZ の上限 IZMM を IY にしたりする。NN は計算される k 一点の番号で 10 節 で説明する関数サブルーチン FUNC は 3 次元の k - 空間の点にこの方法で番号が振られていることを期待している。LATOM(3,MAXLP) に格納されている実格子のセットで与えられている基底関数の中で JPMAT(MAXKPT) に登録されているものだけが、有効であることを注意しながら、DO 32 のループで基底関数に含まれている各格子ベクトルに対して (23) の I_{kT} の計算とか、CRK のデータの選択をして、それらの総和から基底関数値 WW を計算する。これに B(MAXKPT,MAXEIG,2) に格納されている FNNMAT で定めた係数を掛けて k でのエネルギー値が計算される。結果は 7.1 節で述べた全体の見出し P,N,JS に続いて、各スピン、バンド毎に、最小公倍数 N64、 k 点の数 NN、スピンの向き ISPIN、バンドの番号 NB の見出しにつづいて fort.21 に出力される。

8 六方格子




図 1: 六方格子と底心直方格子

この節では、表 1 に含まれている六方格子を、底心直方格子として扱う方法を説明する。この方法を適用すれば、10 節で説明するように、フェルミ面を作図するとき使う関数サブルーチン FUNC に直方、正方、立方格子と同じものが使用できる利点がある。図 1 のように直方格子の b 軸を a 軸に垂直に $\sqrt{3}a$ の長さにとる。つまり b 軸の長さが a 軸の長さのちょうど $\sqrt{3}$ 倍の底心直方格子は六方格子である。




図 2: 逆格子での六方格子と底心直方格子

六方格子と底心直方格子の対応を逆格子で見たのが、図 2 である。六方格子では a^* 軸が a 軸に 30 度傾いて、その大きさが

$$|a_{hex}^*| = \frac{2}{\sqrt{3}|a|} \quad (25)$$

になるのに対し、直方格子では a^* 軸が a 軸の方向で、大きさが

$$|a_{ort}^*| = \frac{1}{|a|} \quad (26)$$

となる。したがって長さの関係は

$$|a_{hex}^*| = \frac{2}{\sqrt{3}}|a_{ort}^*| \quad (27)$$

となり、たがいに 30 度傾いている。また

$$|b_{ort}^*| = \frac{1}{\sqrt{3}}|a_{ort}^*| \quad (28)$$

である。

D_{6h} の場合の既約なブリルアンゾーンは図 2 に多く見えている直角三角形で、赤い矩形で示した領域は 6 倍の体積である。 D_{3d} では、TSP:p67~68 に説明してあるように $P\bar{3}1m$ と $P\bar{3}m1$ の 2 種類があるが、赤い矩形で示した領域は共に 3 倍の体積で既約なブリルアンゾーンをカバーしている。

フェルミ面を作図するとき使う関数サブルーチン FUNC に渡す関数値は、直方格子の a_{ort}^*, b_{ort}^* を NX, NY 個に刻んだ点の関数値であるが、(7) の k は a_{hex}^*, b_{hex}^* を基準にしたものである。両者の変換は同じ k を両方の基準で表して、図 2 から関係を求めると

$$k = xa_{ort}^* + yb_{ort}^* = Xa_{hex}^* + Yb_{hex}^* \quad (29)$$

$$X = (x - y)/2 \quad (30)$$

$$Y = (x + y)/2 \quad (31)$$

となる。プログラム 8.1 はサブルーチン FUEINT からの関係部分の抜粋である。IZ の部分に *2 としてあるのは、(30)、(31) にある /2 を分母の KRMESSH に含めてあるが、これは k_z には余分になるので、ここで調整している。

プログラム 8.1

```
IF(IL.GE.1) THEN
  IRK=LATOM(1,I)*IX*MPX+LATOM(2,I)*IY*MPY
& +LATOM(3,I)*IZ*MPZ
ELSE IF(IL.EQ.0) THEN
  IRK=LATOM(1,I)*(IX*MPX-IY*MPY)
& +LATOM(2,I)*(IX*MPX+IY*MPY)
& +LATOM(3,I)*IZ*MPZ*2
END IF
```

9 詳細な刻みのエネルギーを用いた状態密度:SDOSDR

ほとんどのバンド計算のプログラムは、フェルミエネルギーを出力としてあたえているが、7 節で FUEINT がより詳細な刻みでエネルギーがあたえているので、あらためてフェルミエネルギーを決定し直す必要がある。このために SDOSDR というプログラムが用意されている。四面体法で各バンド毎の状態密度とそのエネルギー積分を計算し、それに基づいてフェルミエネルギー決定する。状態密度とその計算法である四面体法については別に解説を用意しているので、詳しくはそちらに譲ることにして SDOSDR の機能を簡単に述べる。7 節で説明した内挿エネルギーのファイル fort.21 から情報を読み取って計算を実行する。この計算には空間群や格子定数の情報は直接には必要としない。

入力データ 9.1

```
-0.22, 0.0001, 9501, 5.5      EO,DE,NNEE,ELECN
```

入力データ 9.1 は SDOSDR 用の入力データの Cu の例で E0 から、刻み DE で NNEE 個のエネルギー点で状態密度の計算を実行する。最後の ELECEN はバンドに含まれる電子数であるが、使い方に注意が必要である。すべてのバンドは、単位包あたり 1 個の電子と規格化されている。Cu の場合 6 枚のバンドの内挿結果が fort.21 にあたえられている。この場合常磁性で反転対称がある結晶なので 1 枚のバンドに 2 個の電子を収容する。1 1 個の電子がこのバンドに収容されているので、半分にして 5.5 としてある。

入力データ 9.2

```
0.30D0 ,0.0001D0, 7000, 6.0D0
```

入力データ 9.2 は Ni のためのものである。ここでは 3 番から 6 番のバンドがフェルミ面を持つことが分かっているので、この部分だけの内挿値が fort.21 に格納されている。10 個のバレンス電子から 1 番 2 番のバンドの UP と DOWN の電子の分 4 を引いて 6.0 としてある。

10 逆格子空間の任意の点のエネルギー:FUNC

この節では、11 節で説明するフェルミ面の作図用の TPERSP のプログラムの中に含まれている関数サブルーチン FUNC の中で使われているアルゴリズムを説明する。

プログラム 10.1

```
FUNCTION FUNC(XX)
IMPLICIT REAL*8 (A-H,O-Z)
REAL*8 KB
COMMON/ENRSPL/EM(1900000),PN(3),P(3),N(3),IS
DIMENSION IFAC(3)
DIMENSION A(4),IPER(3,6),FF(2,2,2),F(8),KB(3)
EQUIVALENCE(F(1),FF(1,1,1))
DATA IPER/1,2,3, 1,3,2, 3,1,2, 2,1,3, 2,3,1, 3,2,1/
DATA IFAC/1,2,4/
```

TPERSP の説明の pdf の 4 節の隠線処理のアルゴリズムで説明しているように、FUNC には隠線処理のため繰り返し、不特定の連続した点での値が要求される。TPERSP が使用しているこの隠線処理のアルゴリズムは、現在の観点からすれば、効率的なものとはいえないが、任意の箇所を切り取ることを可能にしているなどの機能は捨て難い。現在の計算機の能力が、この非効率さを十分にカバーしている。

FUNC には繰り返し、しかも接近した点での値を要求されることに対応した工夫が盛り込まれている。しかし 20 年以上も前に作られたものであるため、当時の事情を反映しており、現在では少し分かりづらいものになっている。以下に筆者自身が昔の自作のプログラムを解説した結果を紹介する。プログラム 10.1 は FUNC の最初にある宣言文の部分であるが、ここに表れている IPER(3,6) と IFAC(3) の整数データの意味の説明に注意されたい。

10.1 四面体利用線型内挿

アルゴリズムの基本は、図 3 に示された四面体の 4 個の角の値から

$$E = A_1x + A_2y + A_3z + A_4 \quad (32)$$

の一次式の係数を決定して、四面体内部の任意の点での関数値を求める線型内挿である。ここで x, y, z の原点は 1 番の角にとることにする。したがって、 A_4 は 1 番の角の関数値そのものである。

図 3 の直方体は、7 節で説明した FUEINT が計算して FUNC に値をわたした、細かいメッシュの一つを現している。辺の数字は四面体につけた番号で、直方体の角には小さめの数字で示した番号がついてい




図 3: 直方体の 6 個の四面体への分割。

る。図 3 の 6 個の四面体は、どれも 1 番の角と 8 番の角を結ぶ体対角線を、共通の辺として持っている。1 番の角と 8 番の角はすべての四面体で共通である。さらに面の対角線は、各四面体が 2 本ずつ持っているが、隣り合う二つの四面体が共有している。四面体の辺の残り 3 本は直方体の辺である。

図 3 で赤い線で示した 1 番の四面体を見てみよう。ここでの直方体の辺に沿って 1 番の角から 8 番の角へ向かう道筋は、最初 x 方向で、次が y 方向で、最後が z 方向である。この y 方向へ向かう辺はこの四面体だけが占有している。この四面体での、(32) の係数 A_1 は 2 番の角の関数値 E_2 から 1 番の角の関数値 E_1 を引いたもので決まる。両方で y 座標と z 座標が共通であるからである。同様に A_2 は 4 番の角の関数値 E_4 から 2 番の角の関数値 E_2 を引いたもので決まる。また A_3 は 8 番の角の関数値 E_8 から 4 番の角の関数値 E_4 を引いたもので決まる。

ここで、プログラム 10.1 の IPER の数値と図 3 の四面体の番号を見比べてみよう。IPER の数値 1,2,3 を x,y,z と対応させると、6 組の IPER が図の四面体の番号と対応していることが見て取れる。1 番の四面体は上に説明したように、 x,y,z つまり 1,2,3 で、2 番は、 $x,z,y(1,3,2)$ である。以下同様であるが念のため、たとえば 5 番をみてもみると、 $y,z,x(2,3,1)$ となっている。IFAC(3)/1,2,4/ はそれぞれ、 x,y,z 方向に移動したときに角の番号の変化に対応している。つまり x 方向では 1 番だけ変わり、 y 方向ではちょうど 2 番だけ、 z 方向ではいつも 4 番だけ変わっている。

プログラム 10.2

```
C *** DETERMINE THE COEFFICIENTS OF LINEAR FORM ***
K1=IPER(1,IP)
K2=IPER(2,IP)
K3=IPER(3,IP)
I1=IFAC(K1)
I2=IFAC(K2)
I3=IFAC(K3)
A(4)=F(1)
JA=I1+1
```



```

A(K1)=(F(JA)-F(1))/P(K1)
JB=I1+I2+1
A(K2)=(F(JB)-F(JA))/P(K2)
A(K3)=(F(8)-F(JB))/P(K3)

```

プログラム 10.2 の IP が x,y,z の大小関係から決めた四面体の番号である。ここで、I3 を求めているが使用していない。最後はどの四面体でも F(8) になると決まっているからである。

10.2 計算されている領域への座標変換

逆格子空間の中で 7.1 節で説明した領域の中だけで、関数値が用意されているので、逆格子空間の中の任意の点を、この領域内に系の対称性に即して変換することが FUNC の最初の手続きになる。プログラム 10.3 がこの部分にあたっている。XX(3) は関数値が要求された座標で、これを KB(3) に置き換えて変換にとりかかる。もちろん、この KB は REAL*8 KB と宣言してあって、実数変数である。DO 1 のループで D_{2h} の場合の変換が完了する。7.1 節で、P(3) としていた変数がここでは PN(3) になっている。PN(1) は単純格子では 0.5 で、他の格子では 1.0 になっている。つまり、今の座標系は、直交座標ではあるが、その単位は三方向でそれぞれ a^*, b^*, c^* となっている。

このバージョンでは、 C_{4h} の場合の手当がしていないが、(9)、(10)、(11) であたえられている変換の手続きをこの部分に加えればよいことになる。また (17)、(18) の変換をこの部分に加えれば底心直方格子での関数値を用意する範囲を半分にすることができる。

7 節で JS としていた変数がここでは IS となっている。IS=1 なら GO TO 4 となっている。プログラム 10.3 の最後の行は、プログラム 10.4 の先頭と同じ行である。これで座標変換の手続きは終了して、3 図で示されている、どの直方体の中にいるかを見つける次の手続きに移る。IS=2、IS=3 なら KB(3) の 3 個の値の大小関係をみて、データを入れ替える変換をしている。IS=3 で KB(2) と KB(3) を入れ替えたなら、KB(1) と KB(2) の大小関係を再度チェックする必要がある。

プログラム 10.3

```

DO 1 I=1,3
  KB(I)=XX(I)
2  KB(I)=ABS(KB(I))
  IF(KB(I).LE.PN(I)) GO TO 1
  KB(I)=KB(I)-PN(I)*2.0
  GO TO 2
1  CONTINUE
  IF(IS.EQ.1) GO TO 4
C  *** KB(1).GE.KB(2).GE.KB(3) ***
  IF(KB(1).GE.KB(2)) GO TO 3
  W=KB(1)
  KB(1)=KB(2)
  KB(2)=W
3  CONTINUE
  IF(IS.EQ.2) GO TO 4
  IF(KB(2).GE.KB(3)) GO TO 4
  W=KB(2)
  KB(2)=KB(3)
  KB(3)=W
  IF(KB(1).GE.KB(2)) GO TO 4
  W=KB(1)
  KB(1)=KB(2)
  KB(2)=W
4  IX=KB(1)/P(1)

```

10.3 直方体の角の関数値を決定する

プログラム 10.3 で、関数値が用意された領域に送り込まれた KB(3) の点がどこにあるかを次にみつけることになる。この部分がプログラム 10.4 に示されている。まず KB(3) を三つの方向の刻み巾 P(3) で割って整数値 IX、IY、IZ を求める。これらの値がちょうど端の値 N(3) になっているときは、一つ手前にもどしておく。

プログラム 10.4

```
4 IX=KB(1)/P(1)
  IF (IX.EQ.N(1)) IX=N(1)-1
  IY=KB(2)/P(2)
  IF (IY.EQ.N(2)) IY=N(2)-1
  IZ=KB(3)/P(3)
  IF (IZ.EQ.N(3)) IZ=N(3)-1
  IF (IS.EQ.1) NI=(IX*(N(2)+1)+IY)*(N(3)+1)+IZ+1
  IF (IS.EQ.2) NI=((IX*(IX+1))/2+IY)*(N(3)+1)+IZ+1
  IF (IS.EQ.3) NI=(IX*(IX+1)*(IX+2))/6
1  +(IY*(IY+1))/2+IZ+1
C   *** IF THE BLOCK IS THE SAME, GO TO 40 ***
  IF (NI.EQ.NIOLD) GO TO 40
C   *** PICK UP THE ENERGIES OF 8 POINTS ***
  NK=NI
  IF (IS.EQ.1) MX=(N(2)+1)*(N(3)+1)
  IF (IS.EQ.2) MX=(IX+1)*(N(3)+1)
  IF (IS.EQ.3) MX=((IX+1)*(IX+2))/2
  DO 41 KX=1,2
  DO 42 KY=1,2
  DO 43 KZ=1,2
  FF(KX,KY,KZ)=EM(NK)
  NK=NK+1
43 CONTINUE
  IF (IS.NE.3) NK=NK-2+(N(3)+1)
  IF (IS.EQ.3) NK=NK-2+IY+KY
42 CONTINUE
  IF (KX.LT.2) NK=NI+MX
41 CONTINUE
40 CONTINUE
```

関数値が格納されている EM(NI) は添字が一つの配列であるので、IX、IY、IZ から NI を求めなければならない。IS=1,2,3 のそれぞれについて、IX、IY、IZ から NI を求めている式は、7節で説明した、サブルーチン FUEINT の DO 21 IX=,DO 22 IY=,DO 23 IZ= の順序と JS (この FUNC では IS) の扱い方にしたがって、周知の数列の和の公式から簡単に導くことができる。この NI が図 3 の 1 番の角に相当する。次に IF(NI.EQ.NIOLD) GO TO 40 としてあるのが FUNC が接近した座標値にたいして値が要求されることに対応した、応答を速くする一つの工夫である。図 3 の他の7つ角の NK(NI) を求めるのは、それぞれの IS に対する式で IX,IY,IZ の 1 だけの違いが NI にどのように影響するかをみることで簡単にもとまる。結果を FF(KX,KY,KZ)=EM(NK) と格納して、プログラム 10.2 で F(1~8) で使用している。この使い分けのために、プログラム 10.1 で EQUIVALENCE(F(1),FF(1,1,1)) と宣言して二つの配列に同じアドレスをあてがってある。FF の KX,KY,KZ の順序が 3 図の角の番号をあたえている。これは配列内のアドレスに対する FORTRAN の規約にしたがっている。このプログラムで、たとえば IS=2 のとき、IX=IY の直方体の場合に 3 番と 7 番の角の NK が不自然な値になり、したがってこの角の関数値がおかしな値になるがこれは気にしなくてよい。この関数値が使われる 4 番、5 番、6 番の四面体が選ばれることがないからである。

10.4 四面体の番号の決定

プログラム 10.5

```

C   *** FIND THE TETRAHEDRON ***
X=KB(1)-P(1)*FLOAT(IX)
Y=KB(2)-P(2)*FLOAT(IY)
Z=KB(3)-P(3)*FLOAT(IZ)
IF(X.GE.Y) GO TO 11
IF(Y.GE.Z) GO TO 12
C   WRITE(6,*) X,Y,Z
   IP=6
   GO TO 13
12 IF(X.GE.Z) GO TO 24
   IP=5
   GO TO 13
24 IP=4
   GO TO 13
11 IF(Y.GE.Z) GO TO 21
   IF(X.GE.Z) GO TO 22
   IP=3
   GO TO 13
22 IP=2
   GO TO 13
21 IP=1
C   *** IF THE TETRAHEDRON IS THE SAME, GO TO 10 ***
13 IF(NIOLD.NE.NI) GO TO 14
   IF(IPOLD.EQ.IP) GO TO 10
14 IPOLD=IP
   NIOLD=NI

```

つぎに、プログラム 10.5 で 3 図の 1 番の角が原点になるようにして、X,Y,Z を求めてそれらの大小関係から、四面体の番号 IP を決定して、順序を変えて最初に説明したプログラム 10.2 にはいる。

IF(IPOLD.EQ.IP) GO TO 10 で係数の決定をとばしているのは、やはり応答を速くするための工夫である。処理系によっては、これに関連した変数 A,F,IPER,IFAC の SAVE 宣言が必要になる。

11 フェルミ面の実例

11.1 フェルミ面作図プログラム fermhcs.f

この節では 7 節、8 節で説明した FUEINT を含んだ TSBDFT の出力 fort.21 を用いて、立方、正方、直方、六方格子に共通に使える TPERSP のフェルミ面作図プログラム fermhcs.f を紹介する。このプログラムではブリルアンゾーンの外側を切り取るほか、フェルミ面の手前上方の 8 分の 1 を切り取っている。このプログラムは TPERSP の説明の 6 節で説明しているものとはほとんど同じである。以前のものが面心立方格子の Cu の 6 番目のバンドに特化しているのに対し、このプログラムは六方格子を含む、広範な物質に対応している。以下は変更点を中心にした説明である。なお六方格子関連の部分はプログラムでは IF(IL.EQ.0) と明確に読み取れる。説明を分かりやすくするため 11.5 節の PtBi の項で行う。

入力データ 11.1

```

6,1      NB,ISPIN
80.0     SIZE
1        IEQH 1,electron -1 hole
0.4157   EF
129 129 129 1 1 1   NN,NK
15.0     ALPHA

```

各種の状況に対応するため、標準入力からの入力が必要になる。入力データ 11.1 は図 4 のためのデータである。横につけたコメント文から内容は読み取れる。TPERSP の引数から ALPHA(AF) だけを外から変更できるようにしてあるのは、この値が図の見易さに大きく影響するからである。

図の範囲と大きさを決める TPERSP の引数 XFU と WF をプログラム 11.1 のようにしている。これで a^* , b^* , c^* の大きさをそれぞれの方向の単位にし、図の寸法も実際の比になっている。つまり直方格子や、

正方格子では3方向の倍率が異なっている。このバージョンでは扱っていない単斜格子や三斜格子を含めて扱うためには、TPERSPの説明の2節で述べているように3方向とも a^* を単位にした直交座標を採用してTSPACEのサブルーチンRECTAGとGETKVCを活用するのがよいかもしれない。

プログラム 11.1

```

XFU(1,1)=-PN(1)
XFU(2,1)=-PN(2)
XFU(3,1)=-PN(3)
XFU(1,2)=PN(1)
XFU(2,2)=PN(2)
XFU(3,2)=PN(3)
READ(5,*) SIZE
WASIZE=XFU(1,2)-XFU(1,1)
WF(1)=SIZE
WF(2)=SIZE*(A/B)*(XFU(2,2)-XFU(2,1))/WASIZE
WF(3)=SIZE*(A/C)*(XFU(3,2)-XFU(3,1))/WASIZE

```

TSPACEのサブルーチンTSBZEGがあたえるブリルアンゾーンエッジの座標は3方向で共通に a^* を単位にしているので、単位の換算を行っている。

ブリルアンゾーンの外側を切り取る図には切り取った切り口の手当が必要である。CuではブリルアンゾーンのL点に切り口があることが自明なので、8個のL点を中心にして、それぞれの方向の面で切った切り口のTPSECTをCALLしていた。しかし一般にはどのブリルアンゾーンの面に切り口があるかは予想がつかないのですべての面でTPSECTをCALLする。TSBGEGの引数NRECPTは面を与えている逆格子ベクトルを a^*, b^*, c^* 単位であたえているので、面の方向をその方向にし、原点との中点を中心にしてTPSECTをCALLすればよいことになる。ただし図の範囲を指定する面と一致することもあるので、TPFXYZの方のこの部分の引数を0にして切り口の図が重ならないようにする必要がある。

11.2 Cu




図 4: Cu のフェルミ面

図4の3個のフェルミ面のは左から、19個の代表点、85個の代表点、489個の代表点からTSBDFTで内挿した結果を用いた図である。注意して見ないと違いが分からないが、(a)図ではほぼ円形であるはずの胴回りの切り口にわずかな波打ちがみられる。(b)と(c)とでは、(b)の下部に少し不自然な凹凸がみられる。この凹凸が(c)にはみられないので、不自然としているが、(c)がなければ判定の基準はない。代表点をどの密度にとるべきかは一般的には決められない。どの程度の細かい構造を議論するのかに依存している。

11.3 Ni

強磁性金属の例として Ni を取り上げる。AYBAND の 4.4 節の Ni のバンド図と見比べながら説明を読んでいただきたい。Ni では fermhcs.f の手前上部 8 分の 1 を切り取る機能をおさえた図を示す。




図 5: Ni 下向きスピンの第3バンド (a) と第4バンド (b) のホール面

図 5 は X 点の周りの小さな二つのホール面である。それぞれ 0.0046 個と、0.0134 個のホールを含んでいる。AYBAND のバンド図では Δ 軸では両者が縮退しているが S 軸と Z 軸では離れている。




図 6: Ni 下向きスピンの第5バンドの電子面

図 6 は下向きスピンの第5バンドのフェルミ面で電子面として描いている。0.524 個の電子を含んでいるのでホール面としたほうが体積が小さくなるが、図に示すように電子面として閉じている。

図 7 は第6バンド二つのスピンの向きの電子面を示している。それぞれ 0.179 個と 0.310 個の電子を含んでいる。AYBAND の図では Δ 軸で複雑な偶然縮退が起きているのが示されているが、ここに示したフェルミ面には、不自然さはあまり現れていない。




図 7: Ni の第 6 バンドの下向きスピンの (a) と上向きスピンの (b) の電子面

11.4 Mo

体心立方格子の例として Mo を取り上げる。遷移金属で 5s 電子 2 個と、4d 電子 4 個の合計 6 個の価電子を持っている。図 8 の (a) が計算されたバンド図である。常磁性で反転対称を持つので、バンド 1 枚あたり 2 個の電子を収容する。




図 8: Mo のバンド図 (a) と第 5 バンドの電子ポケット (b)

図 8 (a) のバンド図には、フェルミエネルギーと交わるバンドに、上から順に青、緑、赤の色がつけてある。この手法はバンド図からフェルミ面の様子を推測するのに有効である。この物質のスピンの軌道相互作用はあまり強くないので、それを入れないバンドの偶然縮退が僅かに解けた結果の急な折れ曲がりが多く見られる。また Δ 軸、 Λ 軸、F 軸では入れないバンドの二重縮退が僅かに解けているが、図のスケールでは分離して見れない。F 軸から Λ 軸にかけて、青の線と緑の線が重なって見えるがじっさいは僅かに分裂している。計算に使われた代表点は Γ 点と H 点の間を 16 分割した 285 点であった。この密度でバンドの急な変化を大体表すことができている。

青の第 5 バンドは、ほとんどフェルミエネルギーの上を走っているが、 Δ 軸の一部で切っている。この部分のフェルミ面が図 8 (b) で含まれる電子数は 0.009 である。




図 9: Mo の第4バンドの電子面 (a) と第3バンドのホール面 (b)

緑の第4バンドは Γ 点の周りと Δ 軸にかけてフェルミエネルギーの下になっている。 Δ 軸では青のバンドとの間で、上に述べた偶然縮退のスピ軌道相互作用による分裂が見えていてここでバンドの乗り移りが起きている。図 9 (a) では、これに対応して Δ 軸のまわりにフェルミ面の膨らみがある。この面に含まれる電子数は 0.191 個である。

赤の第3バンドは H 点と N 点の周りだけでフェルミエネルギーの上に出ている。図 9 (b) はこの部分をホール面として描いている。H 点の周りは図では 6 個に切れているが、これらは集まって一つの正八面体形の閉じた面になる。また N 点の周りは向かい合った面のものがつながって、6 個の楕円面を形成する。これら 2 種類のフェルミ面に含まれるホール数は 0.200 個である。

11.5 PtBi




図 10: PtBi の結晶構造とブリルアンゾーン

六方格子の例として PtBi を取り上げる。図 10 に示す PtBi の結晶は空間群 $D_{6h}^4, P6_3/mmc$ を持っている。したがってスピ軌道相互作用を入れないバンドでは $k_z = c^*/2$ のブリルアンゾーン面で 2 重縮退している。スピ軌道相互作用は R 軸を除いてこの縮退を解くがあまり大きくはない。

8 節で述べたように、底心直方格子としてエネルギー値が与えられているが、TSPACE には六方格子としてデータが蓄えられている。したがって、fermhcs.f には 5 箇所に IF(IL.EQ.0) の IF 分が挿入されて、




図 11: PtBi のバンド図 (a) と第 14 バンドのホール面

六方格子に対応している。

プログラム 11.2

```
IF(IL.EQ.0) B=SQRT(3.0D0)*A
```

プログラム 11.2 が最初の挿入で、これで六方格子を TSPACE の持っている格子定数から切り離して、このプログラムでは $B = \sqrt{3}A$ の底心直方格子として扱うことになる。

次の挿入はブリルアンゾーンを描く部分である。TSBZEG があたえるブリルアンゾーンエッジの座標 CO(3,2,100) は、 a^* を単位にした直行座標であたえられているが、ここでは直交する三方向でそれぞれ、 a^*, b^*, c^* を単位にしているの、六方格子以外ではスケール変換をして TPLINE に渡している。六方格子では CO のデータを TSPACE のサブルーチン GETKVC を使って、いったん六方格子の $a_{hex}^*, b_{hex}^*, c_{hex}^*$ を単位にした斜交座標の値 AK1,BK1,CK1,AK2,BK2,CK2 に変換している。つぎに (30), (31) を逆に解いた式 $x = X + Y, y = X - Y$ で TPLINE に渡す値を求めている。次はブリルアンゾーンの外側を切り取ったときの切り口に TPSECT を使って蓋をしている部分である。TSBZEG の出力の一つ NRECPT(3,30) はブリルアンゾーンを形成する面に対応する逆格子を表している。通常の格子なら今使っている座標系そのものであるから、そのまま面の方向余弦とし、中点を面が通る点にしている。六方格子では a_{hex}^*, b_{hex}^* 系での値を底心直方格子のそれに変換している。

あと 2 箇所は FUNCND と LFUNC の中で、あたえられた点がブリルアンゾーンの中か外かを判定しているところである。TPERSP が使っている底心直方格子の座標を TSPACE が使用している六方格子の座標に (30),(31) を用いて変換をしている。

図 11 (a) の下にある 2 枚のバンドは Bi の 6s バンドで、その上に Pt の 5d バンドが 10 枚見えている。赤色で示した 14 バンドは R 軸では下の 13 バンドと縮退している。このバンドは Γ 点の周りでフェルミエネルギーの上に出ている。図 11 (b) はこの部分をホール面として描いている。この面に囲まれた体積はブリルアンゾーンの体積の 0.024 にあたり 0.048 個のホールを含んでいる。

図 11 (a) でマゼンダ色で示した第 15 バンドは、R 軸で上の第 16 バンドと縮退しているので図では青色にしている。このバンドは Γ 点から Λ 軸の端の A 点まで、さらに R 軸全体の A 点から L 点まで




図 12: PtBi の第 15 バンドのホール面 (a) と第 16 バンドの電子面

がフェルミエネルギーの上になっている。この 15 バンドのフェルミ面をホール面として描いたのが図 12 (a) である。R 軸に沿った 6 本の腕がフェルミ面の 6 回対称性をよく表している。8 節で導入した六方格子を $B = \sqrt{3}A$ の底心直方格子として扱う方法が、よく機能していることを示している。このフェルミ面に囲まれた体積は 0.277 である。

図 11 (a) でシアン色で示した第 16 バンドは、H 点で最も低いエネルギーになり、P 軸は端の K 点までがフェルミエネルギーの下になっている。K 点から T' 軸に沿って M 点に向かってエネルギーが下がって行く。この 16 バンドのフェルミ面を電子面として描いたのが 12 (b) である。 $k_z = c^*/2$ の面上では R 軸で 15 バンド縮退しているだけでなく、 S, S' 軸でも 15 バンドと 16 バンドが接近しているのが図 11 (a) に示されているが、このことは図 11 (a) と図 11 (b) がこの面の上で互いに相手の隙間に広がっていることにも表れている。

ここで使用している fermfcs.f のフェルミ面の手前上方 8 分の 1 切り取る機能が、図 11 で特徴的な二種類の断面を示している。この格子では c^* 面には 60 度の周期があるので、90 度異なる面がちょうど 30 度だけ異なる面に対応することになる。

11.6 Ga

最後に単体の金属 Ga をとりあげる。この物質は TPERSP と AYBAND の両方の解説でとりあげている。結晶構造と空間群の情報は TPERSP を、ブリルアンゾーンとバンド構造は AYBAND を参照されたい。

AYBAND の説明で触れたように、半金属的なバンド構造でのホール面は対称軸からはずれた箇所が存在している。AYBAND で使用した代表点は 115 点であったが、その内 5 点だけがこのバンドでフェルミエネルギーの上であった。これでは明らかに不十分な代表点の数と予想される。そこで代表点数を 322 点にして、NX,NY,NZ を 192,96,192 にして、TSBDFE を使用して 1,815,937 点の詳細メッシュでフェルミエネルギーを、SDOSDR で決め直した結果 AYBAND の図 3、図 4 で示したバンド図とは少し変わって、ホールバンドが B 軸を切るように変わった。この軸は 2 重縮退であるので、すぐ下のバンドも小さいながらフェルミ面を持つことになる。

上の内挿結果を用いたフェルミ面が図 13 の (a) 図である。ブリルアンゾーンが六角柱になっているが、六




図 13: Ga のホール面 代表点 322(a) と 2119(b)

方格子のそれとは異なって6回対称があるわけではない。ブリルアンゾーンの軸や対称点の名称は TSP:P265 の図 A.18 か、AYBAND の図 2 を参照されたい。直方対称を反映して、 $k_x = 0$ の面を対称面にする大きめの塊と、少し小さめの $k_y = 0$ の面を対称面にする塊がある。 $k_z = 0$ の面に小さな楕円面 (e1) と細いネック (n1) が見えている。 $k_z = c^*/2$ のゾーン境界のネック (n2) は B 軸の上にある。もう一つの対称面 $k_y = b^*/2$ の面では少し太いめの膨らみで隣のゾーンと繋がっているが、その手前の対称面から外れたところに細いネック (n3) がある。さらに上下の大きめの塊をつないで $k_z = 0$ の面に細いネック (n1) を持った上下に伸びた瓢箪型の面と大きめの塊との繋がりも細いネック (n4) になっている。

これらの細かい構造が真実かどうかを確かめるために、さらに代表点の数を 2119 にして描いた図が図 13 の (b) 図である。ここでは (e1) が大きくなって、(n1) と合体している。B 軸上のネック (n2) は大体同じである。(n3) は太くなってネックを形成していない。(n4) は (a) 図より細くなっているが、繋がっていることは確認できる。

代表点が多いのだから (b) 図のほうが確からしいとは言えるが、これが真実であるとは結論できない。確かめるためにはさらに代表点を増やしてバンド計算を実行する必要がある。半金属であることを反映してか 322 点で決めた自己無撞着ポテンシャルは 2119 点でも十分に収斂したポテンシャルであった。したがってより大きな代表点を用いての、自己無撞着計算は不要なので実行することは可能である。しかしそれは本稿の目的を超えていると考える。一般に半金属のフェルミ面はこの Ga のように対称軸を外れたところにあることが多く、しかも形が小さいためブリルアンゾーン全体のスケールからみれば、バンドエネルギーの細かな変化がフェルミ面の形に表れる。

AYBAND の図 4 で青で示したバンドの電子面が図 14 である。(b) 図には代表点の数に伴って、(a) 図よりも細かな構造が見えている。T 点の周りに B 軸と H 軸に沿った塊が見えている。図では 4 個に分かれているが、ブリルアンゾーンの面で切られているため、実際は一つにまとまって閉じたフェルミ面になる。 Σ 軸の周りにあるフェルミ面がブリルアンゾーンの面で切られた切れ端がブリルアンゾーンの縦のエッジが C 軸と交わる場所に見えている。この面は右と左に同じものがあるが、互いに繋がっている訳ではない。

AYBAND の図 4 で緑で示したバンドの電子面が図 15 である。ここでは代表点の数の違いはほとんど見られない。この面は図 14 と B 軸で繋がっている。楕円面のように見えているが、本当は B 軸との交点の先端は尖っている。




図 14: Ga の電子面 (1) 代表点 322(a) と 2119(b)




図 15: Ga の電子面 (2) 代表点 322(a) と 2119(b)